



Merge branch 'rca059-main-patch-33981' into 'main'

mhe authored 13 hours ago

2c8196bf

Data1.md 50.3 KB

# User defined data types - part 1

## Level of difficulty of this handout

This handout includes material of easy, medium, hard and advanced level. If some of the material feels difficult, it is probably because it is difficult rather than your fault. This means you have to work hard if you want to achieve a high mark in the module, as is the case for all modules.

## Video lectures for this handout

The following videos are also linked at appropriate points of this handout for your convenience.

1. [Introduction, the booleans revisited, isomorphisms, Weekdays and the new Maybe type constructor](#) (35 min)
2. [Type retracts](#) (13 min)
3. [Either and And and pairs](#) (9 min)
4. [Lists revisited](#) (9 min)
5. [Binary trees](#) (12 min)
6. [Directions, addresses and paths in binary trees](#) (15 min)
7. [Traversals in binary trees](#) (10 min)
8. [Inverting traversals](#) (18 min)

Total 2hrs.

## Experimenting with the Haskell code included here

You should experiment with the Haskell code in theses notes in order to achieve full understanding. This means running the code and adding things to it, such as solutions of exercises and puzzles, or your own brilliant ideas.

These lecture notes are in [markdown](#) format including Haskell code. To get the Haskell code out of the markdown code, we can use the program [mdtohs.hs](#) included in the [Resources](#) directory, as follows in a Unix/Linux terminal:

```
$ cat Data1.md | runhaskell ../../Resources/mdtohs.hs > Data1.hs
```

This means "copy the contents of the file `Data1.md` to the standard input of the Haskell program `mdtohs.hs` and store the output of the program in the file `Data1.hs`". This can be equivalently written as

```
$ runhaskell ../../Resources/mdtohs.hs < Data1.md > Data1.hs
```

This removes all the markdown code and keeps only the Haskell code, so that we can work with it.

We have already run this for you, and the file [Data1.hs](#) is available in this GitLab repository. Make your own **copy** of this file to avoid conflicts when we update it.

Please note, if you are planning on running your code in GHCi, then you need to first expose the `random` package by running `:set -package random` before you use `:l Data1.hs` to load your file.

## Haskell imports in these lecture notes

Any needed library imports should be mentioned here at the top of the file. We need the following for generating random inputs for testing:

```
module Data1 where

import System.Random
```

## Contents

- [Type synonyms](#)
- [User defined data types](#)

- [The booleans revisited](#)
- [Type isomorphisms](#)
- [Weekdays](#)
- [Some important type constructors](#)
  - [The `Maybe` type constructor](#)
  - [Type `retracts`](#)
  - [The `Either` type constructor](#)
  - [The `And` type constructor, defined by ourselves](#)
- [Lists revisited](#)
  - [Implementing some basic operations on lists](#)
  - [An aside on accumulators](#)
- [Binary trees](#)
  - [Basic functions on binary trees](#)
  - [Directions, addresses and paths in binary trees](#)
  - [Proofs on binary trees by induction](#)
  - [Traversals in binary trees](#)
  - [Inverting traversals \(generating trees\)](#)

## Type synonyms

A video discussing the next few sections is [available on Canvas](#).

Sometimes, mainly for the sake of clarity, we may wish to give a new name to an existing type. For example, the Haskell prelude defines a string to be a list of characters:

```
type String = [Char]
```

Since `String` is just a type synonym, operations such as list concatenation and reverse

```
(++) :: [a] -> [a] -> [a]
reverse :: [a] -> [a]
```

can be freely applied to strings:

```
> "abc" ++ reverse "def"
"abcfed"
```

Type synonyms can also have parameters, as in e.g.

```
type Lst a = [a]
```

## User defined data types

### The booleans revisited

The booleans are defined as follows in Haskell, in the prelude:

```
data Bool = False | True
```

This defines a new type, called `Bool`, with two elements (or *constructors*), called `False` and `True`:

```
False :: Bool
True  :: Bool
```

Functions over a data type can be conveniently defined by **pattern-matching** on its constructors. For example, in the prelude, the conjunction operation

```
(&&) :: Bool -> Bool -> Bool
```

is defined as follows:

```
False && _ = False
```

```
True && x = x
```

A slightly subtle aspect of the semantics of pattern-matching in Haskell is that:

1. the different pattern-matching clauses are tried in order from top to bottom, and
2. the input arguments of the function are only evaluated to the extent needed to check whether they match the current pattern.

A consequence of this semantics is that the above definition of conjunction implements [short-circuit evaluation](#): if the first argument is `False`, then the function returns `False` without even evaluating the second argument.

In contrast, consider the following alternative definition of conjunction:

```
conj :: Bool -> Bool -> Bool
conj False False = False
conj False True  = False
conj True  False = False
conj True  True  = True
```

This version does *not* implement short-circuit evaluation: the second argument will always be evaluated regardless of the value of the first. We can observe the difference between these two versions of conjunction by running the following experiment in the GHC interpreter:

```
> False && undefined
False
> False `conj` undefined
*** Exception: Prelude.undefined
#CallStack (from HasCallStack):
  error, called at libraries/base/GHC/Err.hs:79:14 in base:GHC.Err
  undefined, called at <interactive>:28:11 in interactive:Ghci5
```

## Type isomorphisms

Let's introduce another data type `BW` defined by

```
data BW = Black | White
```

This type is *isomorphic* to the type `Bool`, via the type-conversion functions

```
bw2bool :: BW -> Bool
bw2bool Black = False
bw2bool White = True

bool2bw :: Bool -> BW
bool2bw False = Black
bool2bw True  = White
```

That the pair of functions `(bw2bool, bool2bw)` is an isomorphism means that they are mutually inverse, in the sense that

```
bw2bool(bool2bw b) = b
```

for all `b :: Bool`, and

```
bool2bw(bw2bool c) = c
```

for all `c :: BW`.

Type isomorphisms should *not* be confused with type synonyms. For example, if we try to directly use a value of type `BW` where a value of type `Bool` is expected, we get a type error:

```
> let test = Black && True

<interactive>:39:1: error:
  • Couldn't match expected type 'Bool' with actual type 'BW'
  • In the first argument of '(&&)', namely 'Black'
    In the expression: Black && True
    In an equation for 'it': it = Black && True
```

On the other hand, if we wrap up the values using the explicit coercions `bw2bool` and `bool2bw`, then everything checks:

```
> let test = bool2bw (bw2bool Black && True)
```

Of course, the names `Black` and `White` are arbitrary, and there is another isomorphism between `BW` and `Bool` that swaps `Black` with `True` and `White` with `False` instead.

```
bw2bool' :: BW -> Bool
bw2bool' Black = True
bw2bool' White = False

bool2bw' :: Bool -> BW
bool2bw' False = White
bool2bw' True  = Black
```

And both of the types `Bool` and `BW` are of course isomorphic (again in two different ways each) to the type

```
data Bit = Zero | One
```

of binary digits. One of the isomorphisms is the following:

```
bit2Bool :: Bit -> Bool
bool2Bit  :: Bool -> Bit

bit2Bool Zero  = False
bit2Bool One   = True

bool2Bit False = Zero
bool2Bit True  = One
```

Another one is the following:

```
bit2Bool' :: Bit -> Bool
bool2Bit'  :: Bool -> Bit

bit2Bool' Zero  = True
bit2Bool' One   = False

bool2Bit' False = One
bool2Bit' True  = Zero
```

**Note:** The syntax rules of Haskell require that both type names (here `Bool`, `BW`, `Bit`) and constructor names (here `False`, `True`, `Black`, `White`, `Zero`, `One`) should start with a capital letter.

## Weekdays

Another example of a data type is

```
data WeekDay = Mon | Tue | Wed | Thu | Fri | Sat | Sun
```

We can ask Haskell to do some jobs for free for us (there are alternative ways of doing them ourselves with our own sweat, using type class instances, which we will discuss later):

```
data WeekDay = Mon | Tue | Wed | Thu | Fri | Sat | Sun
    deriving (Show, Read, Eq, Ord, Enum)
```

This automatically adds the type `WeekDay` to the type classes with these five names, which give functions

```
show :: WeekDay -> String
read :: String -> WeekDay
(==) :: WeekDay -> WeekDay -> Bool
(<), (>), (<=), (>=) :: WeekDay -> WeekDay -> Bool
succ, pred :: WeekDay -> WeekDay
```

Look this up in our adopted textbook. Notice that `show` is the counterpart of Java's `toString`, and `read` does the opposite. Some examples are:

```
> show Tue
"Tue"
```

```
> read "Tue" :: WeekDay -- (the type annotation tells Haskell to try to parse the string as a WeekDay)
Tue
> read "Dog" :: WeekDay
*** Exception: Prelude.read: no parse
> Mon == Tue
False
> Mon < Tue
True
> succ Mon
Tue
> pred Tue
Mon
> [Mon .. Fri]
[Mon,Tue,Wed,Thu,Fri]
```

Monday doesn't have a predecessor, and Sunday doesn't have a successor:

```
> pred Mon
*** Exception: pred{WeekDay}: tried to take `pred' of first tag in enumeration
CallStack (from HasCallStack):
  error, called at Data1.hs:20:47 in main:Main
> succ Sun
*** Exception: succ{WeekDay}: tried to take `succ' of last tag in enumeration
CallStack (from HasCallStack):
  error, called at Data1.hs:20:47 in main:Main
```

Notice that the choice of names in the type of weekdays is arbitrary. An equally good, isomorphic definition is

```
data WeekDay' = Monday | Tuesday | Wednesday | Thursday | Friday | Saturday | Sunday
```

## Some important type constructors

### The `Maybe` type constructor

Sometimes a function may not be able to give a result, in which case we would like it to say explicitly that it cannot. We use the `Maybe` type from the prelude for that purpose:

```
data Maybe a = Nothing | Just a
```

Here `a` is a type parameter, and we have the following types for `Nothing` and `Just`:

```
Nothing :: Maybe a
Just    :: a -> Maybe a
```

This means that the constructor `Just` is a function. It [converts](#) an element of type `a` into an element of type `Maybe a`. So this function `Just` is a so-called *type coercion*, also known as a *type cast*.

For example:

```
Just 17 :: Maybe Integer
```

In summary, the only possible elements of the type `Maybe a` are `Nothing` and `Just x` where `x` has type `a`.

In Java the `Maybe` type constructor is called `Optional`.

### Example: integer computations that may give errors

In the following definition of division, if the denominator is zero, then division is impossible, and so the result is `Nothing`. If it is possible, we simply perform the division, and we convert the resulting `Int` to `Maybe Int` using the type conversion function `Just`, which then gives the result of the function `divide`:

```
divide :: Int -> Int -> Maybe Int
x `divide` y = if y == 0 then Nothing else Just (x `div` y)
```

For example, we get

```
>10 `divide` 2
```

```
Just 5
> 10 `div` 0
Nothing
```

But now suppose you want to do `3 + (10 `div` 0)`. You would expect `Nothing` but instead this expression doesn't even type check:

```
> 3 + (10 `div` 0)

<interactive>:11:1: error:
• No instance for (Num (Maybe Int)) arising from a use of `+`
• In the expression: 3 + (5 `div` 0)
  In an equation for `it`: it = 3 + (5 `div` 0)
```

What this is saying is that `(10 `div` 0)` is expected to be an `Int`, because `+` expects an `Int` as its right argument, but it isn't. That's because it is a `Maybe Int`. So we need a version of addition that can cope with errors as its possible inputs:

```
adde :: Maybe Int -> Maybe Int -> Maybe Int
adde Nothing Nothing = Nothing
adde Nothing (Just y) = Nothing
adde (Just x) Nothing = Nothing
adde (Just x) (Just y) = Just (x + y)
```

Now to fix `3 + (5 `div` 0)` we replace `+` by `adde`, but also we need to convert the number `3` to the type `Maybe Int` with the *type coercion* or *type cast* `Just`.

```
> Just 3 `adde` (10 `div` 2)
Just 8
> Just 3 `adde` (10 `div` 0)
Nothing
```

A more concise definition of `adde`:

```
adde' :: Maybe Int -> Maybe Int -> Maybe Int
adde' (Just x) (Just y) = Just (x+y)
adde' _ _ = Nothing
```

This works because the execution of Haskell programs tries patterns from top to bottom, and the last pattern catches all remaining possibilities.

A definition using `cases` is also possible:

```
adde'' :: Maybe Int -> Maybe Int -> Maybe Int
adde'' xm ym = case xm of
  Nothing -> Nothing
  Just x -> case ym of
    Nothing -> Nothing
    Just y -> Just (x+y)
```

Later we will see that there is a much more concise way of making such definitions using *monads*. But for now we will stick to pattern matching and cases.

### Example: find the first position an element occurs in a list

Since the first position is undefined if the element doesn't occur in the list, in that case we answer `Nothing`:

```
firstPosition :: Eq a => a -> [a] -> Maybe Integer
firstPosition x [] = Nothing
firstPosition x (y:ys)
  | x == y = Just 0
  | otherwise = case firstPosition x ys of
    Nothing -> Nothing
    Just n -> Just(n+1)
```

For example:

```
> firstPosition 'a' ['a'..'z']
Just 0
> firstPosition 'b' ['a'..'z']
Just 1
```

```
> firstPosition 'z' ['a'..'z']
Just 25
> firstPosition '!' ['a'..'z']
Nothing
```

which we summarize as

```
firstPosition 'a' ['a'..'z'] = Just 0
firstPosition 'b' ['a'..'z'] = Just 1
firstPosition 'z' ['a'..'z'] = Just 25
firstPosition '!' ['a'..'z'] = Nothing
```

You are required to use the book to find out what `case` is and how it works in general, but in this example it should be clear. You are also required to use the book to find out about conditional definitions using `|` to indicate *guards* for equations.

We will use the `Maybe` type constructor very often, because there are many occasions in which some inputs are *invalid*.

*Task.* Define a function `allPositions :: Eq a => a -> [a] -> [Int]` that finds all positions in which an element occurs in a list. For example, we should have `allPositions 17 [13,17,17,666] = [1,2]` and `allPositions 17 [1,2,3] = []`.

## Type retracts

A [video](#) discussing this section is available.

This section may be rather hard at your current level. Consider it as a challenge. If you get lost, feel free to skip it at a first reading, go to the next section, and come back to it later. This is important because it deals with data coding, which is an important, or even crucial, aspect of Computer Science. What we say here is in Haskell, but it actually applies to any programming language.

We have already seen examples of type *isomorphisms*. For example, the type `Bool` is isomorphic to the type `BW` of black-and-white colours, and also to the type `Bit` of binary digits `Zero` and `One`.

More generally, an isomorphism of types `a` and `b` is a pair of functions

```
f :: a -> b
g :: b -> a
```

such that

- `f (g y) = y` for all `y :: b`, and
- `g (f x) = x` for all `x :: a`.

We summarize these two equations by saying that these two functions are *mutually inverse*. This means that we can convert back and forth between elements of the type `a` and elements of the type `b`, like we did when we converted `False` to `Zero` and `True` to `One`. You can think of `f` and `g` as *renaming* functions: the function `f = bit2Bool` renames a bit to a boolean, and the function `g = bool2Bit` renames a boolean to a bit.

In practice, this means that it doesn't matter whether we work with the type `Bool` with elements `True` and `False`, or the type `Bit` with elements `Zero` and `One`. In fact, computers work by exploiting this identification of booleans with binary digits.

There is another relationship between types that is also useful in practice: a type `b` can "live" inside another type `a`, in the sense of having a "copy" in the type `a`. A simple example is that the type `Bool` has a copy inside the type `Int`:

```
bool2Int :: Bool -> Int
bool2Int False = 0
bool2Int True  = 1
```

Not only do we have a copy of `Bool` inside `Int`, but also we can go back, so that we get `False` and `True` from `0` and `1`:

```
int2Bool :: Int -> Bool
int2Bool n | n == 0  = False
           | otherwise = True
```

However, notice that not only `1` is converted back to `True`, but also everything other than `0` is converted to `True`.

We have

```
int2Bool (bool2Int y) = y
```

for every `y :: Bool`, but we don't have `bool2Int (int2Bool x) = x` for all `x :: Int`, as this fails for e.g. `x = 17` because `bool2Int (int2Bool 17)` is `1` rather than `17`.

We can say that there is enough room in the type integers for it to host a copy of the type of booleans, but there isn't enough room in the type of booleans for it to host a copy of the type of integers.

When there are functions

```
f :: a -> b
g :: b -> a
```

such that

- $f (g y) = y$  for all  $y :: b$ ,

but not necessarily  $g (f x) = x$  for all  $x :: a$ , we say that the type  $b$  is a *retract* of the type  $a$

Our discussion above shows that the type `Bool` is a retract of the type `Int`. This retraction is the same as that performed in the programming language `C`, where the integer `0` codes `False` and everything else codes `True`.

But notice that there are other ways in which the type `Bool` lives inside the type `Int` as a retract: for example, we can send `False` to `13` and `True` to `17`, and then send back everything bigger than `15` to `True` and everything else to `False`. We are free to code things as we wish.

**Task.** Show that the type `Maybe a` is a retract of the type `[a]`. The idea is that `Nothing` corresponds to the empty list `[]` and that `Just x` corresponds to the one-element list `[x]`. Make this idea precise by writing back and forth functions between these types so that they exhibit `Maybe a` as a retract of `[a]`. Our adopted textbook exploits such a retraction often, albeit without making it explicit. In fact, what the book does very often is to avoid the type `Maybe a` and instead work with the type `[a]`, considering only the list `[]` (corresponding to `Nothing`) and singleton lists `[x]` (corresponding to `Just x`), and ignoring lists of length `2` or greater. (The reason the book does that is to avoid monads (before they are taught) in favour of list comprehensions (which are taught early on), as list comprehensions happen to accomplish the same thing as "do notation" for monads, in the particular case of the list monad. So this coding is done for pedagogical purposes in this case.)

If we have a type retraction  $(f, g)$  as above, then:

- $f$  is a **surjection**.

This means that for every  $y :: b$  there is **at least one**  $x :: a$  with  $f x = y$ .

For example, in the case of the booleans as a retract of the integers, this means that every boolean is coded by at least one integer.

- $g$  is an **injection**.

This means that for every  $x :: a$  there is **at most one**  $y :: b$  with  $g y = x$ .

In the first example of the booleans as a retract of the integers, this is the case:

- For  $x = 0$  we have exactly one  $y$  with `bool2Int y = x`, namely  $y = \text{False}$ .
- For  $x = 1$  we have exactly one  $y$  with `bool2Int y = x`, namely  $y = \text{True}$ .
- For  $x$  different from `0` and `1` we have no  $y$  with `bool2Int y = x`.

So for every  $x$  there is at most one such  $y$  (i.e. exactly one or none).

**Task.** Define

```
data WorkingWeekDay = Mon' | Tue' | Wed' | Thu' | Fri'
```

We add primes to the names because the names without prime are already used as elements of the type `WeekDay` defined above. Show that the type `WorkingWeekDay` is a retract of the type `WeekDay`. Arbitrary choices will need to be performed in one direction, like e.g. the language `C` arbitrarily decides that any integer other than `0` codes `true`.

**Puzzle.** Consider the function

```
g :: Integer -> (Integer -> Bool)
g y = \x -> x == y
```

We can visualize  $g$  in the following table:

	...	-5	-4	...	-1	0	1	...	4	5	...
$g(-5)=$	...	True	False	...	False	False	False	...	False	False	...
$g(-4)=$	...	False	True	...	False	False	False	...	False	False	...
...											
$g(-1)=$	...	False	False	...	True	False	False	...	False	False	...



	...	-5	-4	...	-1	0	1	...	4	5	...
g(0)=	...	False	False	...	False	True	False	...	False	False	...
g(1)=	...	False	False	...	False	False	True	...	False	False	...
...											
g(4)=	...	False	False	...	False	False	False	...	True	False	...
g(5)=	...	False	False	...	False	False	False	...	False	True	...

That is, the function `g` codes the integer `y` as the function `h` such that `h y = True` and `h x = False` for `x` other than `y`. Convince yourself that the function `g` is an injection. In this sense, the type `Integer` lives inside the function type `Integer -> Bool`. Do you think `g` has a companion `f` : `(Integer -> Bool) -> Integer` that "decodes" functions `Integer -> Bool` back to integers, such that for any code `g y` of the integer `y` we get the integer back as `f (g y) = y`? If yes, then give a Haskell definition of such an `f` and convince yourself that indeed `f (g y) = y` for all integers `y`. If not, why? This puzzle is rather tricky, and none of the possible answers "yes" or "no" to the question is obvious.

## The `Either` type constructor

A video discussing the next few sections is [available on Canvas](#).

It is defined in the prelude as follows:

```
data Either a b = Left a | Right b
```

Then we have

```
Left  :: a -> Either a b
Right :: b -> Either a b
```

For example:

```
Left 17      :: Either Integer String
Right "abd"  :: Either Integer String
```

The idea is that the type `Either a b` is the *disjoint union* of the types `a` and `b`, where we tag the elements of `a` with `Left` and those of `b` with `Right` in the union type. An example of its use is given below.

## The `And` type constructor, defined by ourselves

The following has an isomorphic version predefined in the language, as we shall see soon:

```
data And a b = Both a b
```

This is a type constructor with two parameters, and with an element constructor `Both`, which is a function

```
Both :: a -> b -> And a b
```

For example, assuming we have defined types `MainDish`, `Dessert`, `Drink`,

```
data MainDish = Chicken | Pasta | Vegetarian
data Dessert  = Cake | IceCream | Fruit
data Drink    = Tea | Coffee | Beer
```

we can define:

```
type SaverMenu = Either (And MainDish Dessert) (And MainDish Drink)
```

which can be equivalently written

```
type SaverMenu = Either (MainDish `And` Dessert) (MainDish `And` Drink)
```

(Choose which form of definition you like better. Haskell accepts both.)

So what is available in the saver menu is either a main dish and a dessert, or else a main dish and a drink. It should be intuitively clear that this is

isomorphic to

```
type SaverMenu' = And MainDish (Either Dessert Drink)
```

meaning that you can have a main dish and either dessert or a drink. This intuition is made precise by the isomorphism

```
prime :: SaverMenu -> SaverMenu'
prime (Left (Both m d)) = Both m (Left d)
prime (Right(Both m d)) = Both m (Right d)

unprime :: SaverMenu' -> SaverMenu
unprime (Both m (Left d)) = Left (Both m d)
unprime (Both m (Right d)) = Right(Both m d)
```

So, as a software developer, you can choose either `SaverMenu` as your implementation, or else `SaverMenu'`. They are different, but essentially equivalent.

We actually don't need to define `And`, because an equivalent type constructor is already available in Haskell, namely the type of pairs. We have an isomorphism as follows:

```
and2pair :: And a b -> (a,b)
and2pair (Both x y) = (x,y)

pair2and :: (a,b) -> And a b
pair2and (x,y) = Both x y
```

And so we have further isomorphic versions of the saver menu type:

```
type SaverMenu'' = Either (MainDish, Dessert) (MainDish, Drink)
type SaverMenu''' = (MainDish, Either Dessert Drink)
```

Lookup the type of pairs (tuple types) in the book and read about it.

## Lists revisited

A video discussing the next few sections is [available on Canvas](#).

With a pinch of salt, the type of lists is predefined by

```
data [a] = [] | a : [a]  -- not quite a Haskell definition
```

which says that a list of `a`'s is either empty, or else an element of the type `a` followed (indicated by `:`) by a list of `a`'s. This is an example of a *recursive* data type definition. We have the following types for the list constructors:

```
[] :: [a]
(:) :: a -> [a] -> [a]
```

Although the above not-quite-a-Haskell-definition is semantically correct, it is syntactically wrong, because Haskell (unfortunately) doesn't accept this kind of syntactical definition. If we don't care about syntax, we can define an isomorphic version as follows:

```
data List a = Nil | Cons a (List a)
```

Then the types of the constructors are

```
Nil  :: List a
Cons :: a -> List a -> List a
```

For example, the native list `[1,2,3]` is written `Cons 1 (Cons 2 (Cons 3 Nil))` in our isomorphic version of the type of lists. Let's define the isomorphism to make this clear:

```
nativelist2ourlist :: [a] -> List a
nativelist2ourlist []      = Nil
nativelist2ourlist (x:xs) = Cons x (nativelist2ourlist xs)

ourlist2nativelist :: List a -> [a]
ourlist2nativelist Nil      = []
```

```
ourlist2nativelist (Cons x xs) = x:ourlist2nativelist xs
```

Notice that these coercions are defined recursively, corresponding to the fact that the data type itself is defined recursively.

## Implementing some basic operations on lists

Let's write our own versions of the list concatenation ("append") and reverse operations from the prelude:

```
append :: List a -> List a -> List a
append Nil      ys = ys
append (Cons x xs) ys = Cons x (append xs ys)

rev :: List a -> List a
rev Nil        = Nil
rev (Cons x xs) = rev xs `append` (Cons x Nil)
```

We can try to test that these do the right thing by comparing them to the implementations of list concatenation and reversal in the Haskell prelude, using the isomorphism between `List a` and `[a]`. Indeed, we expect that

```
ourlist2nativelist (append (nativelist2ourlist xs) (nativelist2ourlist ys)) == xs ++ ys
```

and

```
ourlist2nativelist (rev (nativelist2ourlist xs)) == reverse xs
```

should evaluate to `True` for all native lists `xs`, `ys :: [a]`. Let's test these properties:

```
> let xs = [1..5]
> let ys = [6..10]
> ourlist2nativelist (append (nativelist2ourlist xs) (nativelist2ourlist ys)) == xs ++ ys
True
> ourlist2nativelist (rev (nativelist2ourlist xs)) == reverse xs
True
```

Of course, here we have only tested on a couple examples, but it is true in general. (Question: how would you *prove* this?)

Although our definitions are functionally correct, there is a more subtle problem with our implementation of `rev`. By inspection of the code, `append xs ys` computes the concatenation of two lists in time  $O(n)$ , where  $n$  is the length of `xs`, since each recursive call to `append` decreases the length of `xs` by one, and the calls to `Cons` are constant time. On the other hand, `rev` is  $O(n^2)$  by the same argument, since each recursive call to `rev` decreases the length of `xs` by one, and each call to `append` is  $O(n)$ .

This is not just a theoretical problem — we quickly bump into it if we compare reversing a reasonably large list using the native `reverse` function versus the implementation `rev` above.

```
> let xs = [1..10^5]
> length (reverse xs) -- this is fast (we return the length of the reversed list in order to keep the output small)
100000
> length (ourlist2nativelist (rev (nativelist2ourlist xs))) -- this is really slow, so we give up and hit Control-C
C-c C-cInterrupted.
```

There's a much more efficient way of implementing reversal by introducing a helper function with an extra argument:

```
fastrev :: List a -> List a
fastrev xs = revapp xs Nil
  where
    revapp :: List a -> List a -> List a
    revapp (Cons x xs) ys = revapp xs (Cons x ys)
    revapp Nil      ys = ys
```

One way to think of the second argument of the helper function `revapp` is as a stack, initially set to be empty (`Nil`). The function recursively scans the input from the first argument, pushing each element onto the stack (second argument). When there are no more input elements, the stack is simply popped directly to the output, with all of the elements of the original list now in reverse order.

Here's a concrete illustration of how this works, unrolling the definitions of `fastrev` and `revapp` to reverse a four-element list:

```
fastrev (Cons 1 (Cons 2 (Cons 3 (Cons 4 Nil))))
= revapp (Cons 1 (Cons 2 (Cons 3 (Cons 4 Nil)))) Nil
= revapp (Cons 2 (Cons 3 (Cons 4 Nil))) (Cons 1 Nil)
```

```

= revapp (Cons 3 (Cons 4 Nil)) (Cons 2 (Cons 1 Nil))
= revapp (Cons 4 Nil) (Cons 3 (Cons 2 (Cons 1 Nil)))
= revapp Nil (Cons 4 (Cons 3 (Cons 2 (Cons 1 Nil))))
= Cons 4 (Cons 3 (Cons 2 (Cons 1 Nil)))

```

Another way of thinking of the function `revapp` is suggested by its name: given two lists `xs` and `ys`, we have that `revapp xs ys` computes the reversal of `xs` appended with `ys`. It's not hard to see that this binary operation `revapp` is *more general* than the original unary reversal operation: the latter can be recovered by taking `ys = Nil`. On the other hand, `revapp` is much more efficient than our original function `rev`, being only  $O(n)$  in the length of its first argument `xs`.

This pattern — where we manage to solve a problem or solve it more efficiently by replacing it with a more general and seemingly more difficult problem — happens again and again in functional programming.

## An aside on accumulators

The extra argument `ys` that we used in the helper function `revapp` is sometimes called an "accumulator", since it accumulates a value that is eventually passed to the output. Above we saw how an accumulator could be used to turn an  $O(n^2)$  algorithm into an  $O(n)$  algorithm for list reversal. For an even starker example, consider the problem of computing the [Fibonacci numbers](#)  $F_n$ .

The mathematical definition of the Fibonacci sequence in Wikipedia may be translated directly into the following Haskell code:

```

fib 0 = 0
fib 1 = 1
fib n = fib (n-1) + fib (n-2)

```

But although this definition is correct, it is extremely inefficient!

We can already see this if we try to use the above definition to compute, say, the first 32 Fibonacci numbers:

```

> :set +s -- ask ghci to print time and space usage
> [fib n | n <- [0..31]]
[0,1,1,2,3,5,8,13,21,34,55,89,144,233,377,610,987,1597,2584,4181,6765,10946,17711,28657,46368,75025,121393,196418,317811,514229,832040,1346269,2178309,3542248,5717580,9271035,14930352,24146880,39088373,63225695,102334155,166557648,268832813,435390461,704147680,1139567641,1844706100,2984072741,4823539841,7807625582,12631565423,20439068799,33071604222,53501360711,86532964960,139034370681,225536177191,364570547872,590106718063,954636895254,1544757062915,2500393680169,4045150643084,6545548113253,10590701913422,17135850076675,27726551970137,44862351983812,72588202053947,117414754034084,190062156088021,307476858142068,497541558176189,805018714318257,1302559273458446,2107578812624635,3410138086083081,5517716898707717,8927854981332352,14442572877420169,23370427868727921,37813000746148080,61183428614875901,99006429383603881,159189858001581781,258196386815185662,417196245298787443,675396123900369124,1092592410115554965,1767988696013744089,2860581106129313213,4628569596244868182,7490551702364182171,12119120898594050353,19619682594743232524,31739803496337282877,51359484397931333221,83079387894270015098,134418871891201346919,217498259888131362140,351917637883401709059,569416519874533071278,921415397762634373418,1490832016646036094657,2412247414410569465975,3903082412172582559593,6315329826608618654568,10218412238781201174161,16533742065389820828729,26752154304171439983890,43285896370560660602619,70038638436742081431548,113291834737302742415447,183330473108044824018066,296622311845347566433513,479914144953350308848959,776244558061398132864472,1256166879916745641302431,2032381027970143774151380,3288547807586941915453811,5320928837503685999656242,8609310065474127314108053,13930238863080813730061895,22539559142854941044718138,36469797907935764354780033,59009036050816578084842471,95438824193752342439600604,154447860144568910494383035,250039368016585488579225506,404487228161344399073608541,654515064308130879572784047,1059070191342256273968392588,1713585007667417153540116635,2772655197013673427508509223,4486235198381090581448625858,7258820205394707735048741493,11741475403408378216557250716,19006215608802085951505992209,30747685814206783767063243025,49754155817618462018571495234,80501871431825245775634738259,130255927345841928003697981283,210757881262463673779332719542,341013808608305591783030690825,551771689870769264762363402367,892785498133230856741696114150,1444257287742046448528059515027,2337042786872516306290422627394,3781300074614832744718582142421,6118342861487549051218999769815,9900642938360365368929422912242,15918985800158134419148422682057,25819638681518293616067422451872,41719624529878588935216245134119,67539612390036882554365067586001,109259241011555173773503292720120,176798869601374062432818359306121,286058110612931751606321581026241,462856959624486924779139938546362,749055170236418676382658219866483,1211912089859406408111791798392844,1961968259474323280790924017459227,3173980349633728296902715816252070,5135948439793133313014507633711297,8307938789427001529126299450172517,13441887189120134481144117083883814,21749825988813136642262035258496081,35191763788340170803379953433109295,56941651987453307004497871607721476,92141539776263437205615789782332771,149083201664603609056794971957944047,241224741441056941067974153713116122,390308241217258253079153335469228393,631532982660861865090332517225344515,1021841223878120117102510734981472908,1653374206538982082714692952737617423,2675215430417143992826484770493790331,4328589637056066062938276588250007754,7003863843674208143050068406006220085,11329183473730274243161860223762437819,18333047310804482404279778401324659674,29662231184534756605397696579187097593,47991414495335030806515614756749257467,77624455806139813007633532934300417340,125616687991674564008811451111859675007,203238102797014376019990632887471272514,328854780758694191031169814664963869921,532092883750368593042348996442455442436,860931006547412733053528178217967052357,1393023886308081353065310000000000000000,2107578812624636730077091817755116052367,3410138086083055910188873635510232151734,5517716898707692630300655453265348251101,8927854981332308550412437271020464350468,14442572877420464301540255048755580449835,23370427868725163102658073226306741444502,37813000746148327303775891403857902439069,61183428614875490504893709581409063433738,99006429383603653706011527758960224428407,159189858001581343017129705536511385423076,258196386815182935028307887312022995369751,417196245298785887039486069087534605315422,675396123900368823050664250863046215261993,1092592410115551730061842028638557825208644,1767988696013740610173623846414069435155355,2860581106129317530285405664169185544622006,4628569596244869250397187481924301644088617,7490551702364186770508969299679417743555128,12119120898594064000626787517429533843022239,19619682594743232801744565695184649937689350,31739803496337282962862383872735816032356461,5135948439793133312398020205028697702702272,8307938789427001532519802022783813802168883,13441887189120134322637618200334974901635494,21749825988813136523755436377886136001202105,35191763788340170724873254555437297095868216,56941651987453306925991072732988458090534327,92141539776263437127108890910539619085199438,149083201664603609260286683088090780080864549,241224741441056941271464864863602391026529660,390308241217258253282643046640114001772194771,631532982660861865293821228415625612717860882,1021841223878120117295000010193137223664521993,165337420653898208290617782796864883461118310,267521543041714399301795964573376494406783421,432858963705606606312974146348888105352448532,700386384367420814324152328124399716308113643,1132918347373027424333330145900911327264778754,1833304731080448240345107963676422937721439865,2966223118453475660456885781431534547178090976,4799141449533503080568663599186650656634742087,7762445580613981300680441416941766766091393198,12561668799167456400792219234696882875548044309,20323810279701437601910037412451998985104695420,32885478075869419103027815590103150080671206531,53209288375036859304145633767654311175237817642,86093100654741273305263451945205472269804328753,139302388630808135306441631722757083364370839864,210757881262463673017619413500308244460036950975,341013808608305591028797595275819855405703062086,551771689870769263039975777051331466351369173197,892785498133230855051153958826843077307034284308,1444257287742046430163935780601354688252700395419,2337042786872516310275713598356470797708361506530,3781300074614832730387495416111586907164022617641,6118342861487549050499277233866703016620683728752,9900642938360365370611059051621819126077344839863,1591898580015813430172281782942293523553400000000,2581963868151829350284063600707405134509056000000,4171962452987858870395845418462521244964712000000,6753961239003688230507627236217637354420372000000,10925924101155517300619445014022753463976982000000,17679886960137406101737263191573914563533642000000,28605811061293175302855081369125075658100302000000,46285695962448692503972899546676236752666962000000,74905517023641867705090717724227397847233562000000,121191208985940640006268695001879008942800162000000,196196825947432328017446476777390620038366762000000,317398034963372829628624658552902231034032962000000,513594843979313331239802836328413842029700562000000,830793878942700153251981018103925453025368562000000,1344188718912013432263802800879341063081034562000000,2174982598881313652375580618644457173036700562000000,3519176378834017072487362636409573283093360562000000,5694165198745330692599144454164689393050020562000000,9214153977626343712710926271919805502606680562000000,14908320166460360926029044489670021612163340562000000,24122474144105694127146862667221182717730000562000000,39030824121725825328264660844872343813296660562000000,63153298266086186529382479022423504918863260562000000,102184122387812011729500256800025666024430260562000000,165337420653898208290618038577537227120096860562000000,267521543041714399301796220354048838165763460562000000,432858963705606606312974402130560449111430060562000000,700386384367420814324152583906072060067096060562000000,1132918347373027424333330261681583671022762060562000000,1833304731080448240345108039457095281578428060562000000,2966223118453475660456885817212211391135094060562000000,4799141449533503080568663634967327500691754060562000000,7762445580613981300680441452722443610248414060562000000,12561668799167456400792219340473604720805074060562000000,20323810279701437601910037518228720830361734060562000000,32885478075869419103027815695779881930928394060562000000,53209288375036859304145633873331043036495054060562000000,86093100654741273305263452050882204132061654060562000000,139302388630808135306441632726333815237628254060562000000,210757881262463673017619414503845426233194854060562000000,341013808608305591028797596779365837238810854060562000000,551771689870769263039975778554877448194476854060562000000,892785498133230855051153960330389059150142854060562000000,1444257287742046430163935783105905170105808854060562000000,2337042786872516310275713600861021280661474854060562000000,3781300074614832730387495418616137390218140854060562000000,6118342861487549050499277236371253500774800854060562000000,9900642938360365370611059054126369610331460854060562000000,15918985800158134301722817831924485720888120854060562000000,25819638681518293502840636009475646826444780854060562000000,41719624529878588703958454189586807932000340854060562000000,67539612390036882305076272367137969037566940854060562000000,109259241011555173006194450167880130143132540854060562000000,176798869601374061017372631943391741198708140854060562000000,286058110612931753028550813718903352254373740854060562000000,462856959624486925039728995494414963310039340854060562000000,749055170236418677050907177270426574355995340854060562000000,1211912089859406400062686950091248185411651340854060562000000,1961968259474323280174464767846364295967307340854060562000000,3173980349633728296286246585601480406023963340854060562000000,5135948439793133312398028363369600516580629340854060562000000,8307938789427001532519810181124716627137289340854060562000000,13441887189120134322638028008807267737693949340854060562000000,21749825988813136523755806186358428843250609340854060562000000,35191763788340170724873626363959589948817269340854060562000000,56941651987453306925991444541510751054383869340854060562000000,92141539776263437127109262719101912159950469340854060562000000,149083201664603609260290444896713523265517069340854060562000000,241224741441056941271468626672225134321183669340854060562000000,390308241217258253282646608448736745376850269340854060562000000,631532982660861865293824790224248356432416269340854060562000000,1021841223878120117295002568000256667488082269340854060562000000,1653374206538982082906180385775374878543748269340854060562000000,2675215430417143993017962203540490989103414269340854060562000000,4328589637056066063129744021305707099660074269340854060562000000,7003863843674208143241525839060823210226734269340854060562000000,11329183473730274243333302616816934320783394269340854060562000000,18333047310804482403451080394572050431340054269340854060562000000,29662231184534756604568858172123661536906714269340854060562000000,47991414495335030805686636349674822642473374269340854060562000000,77624455806139813006804414527225983748040034269340854060562000000,125616687991674564007922193404737034853606634269340854060562000000,203238102797014376019100375182288645909173234269340854060562000000,32885478075869419103027815695780025
```

```

while (n > 1) {
  int z = x+y;
  x = y;
  y = z;
  n = n-1;
}
return (n == 0 ? x : y);
}

```

In addition to this low-level view of what `fastfib` and in turn `fibAcc` is doing, there is also a more high-level view (as we saw before in the case of `revapp`). Can you identify a sense in which the helper function `fibAcc n x y` computes a *more general* function than `fib n`?

## Binary trees

A video discussing the next few sections is [available on Canvas](#).

A binary tree over elements of a type `a` is either empty, or consists of a root labelled by an element of the type `a` followed by two binary trees, called the left and right subtrees:

```
data BT a = Empty | Fork a (BT a) (BT a)
```

- We have the empty tree, to be called `Empty`. We take the convention that empty trees are drawn as dangling leaves.

```
|
```

- Given two trees `l` and `r` and an element `x :: a`, we have the new tree

```

  x
 / \
l   r

```

written `Fork x l r`.

For example, the tree

```

      8
     / \
    /   \
   4     16
  / \   / \
 2   20 /  \
 /   \ /   \

```

is written, in this notation, as

```
btexample = Fork 8 (Fork 4 (Fork 2 Empty Empty) Empty) (Fork 16 Empty (Fork 20 Empty Empty))
```

We can ask Haskell to do some work for us by deriving things as above.

```
data BT a = Empty
          | Fork a (BT a) (BT a) deriving (Show, Read, Eq, Ord)
```

We have that

```


Empty :: BT a
Fork  :: a -> BT a -> BT a -> BT a

```

**Puzzle.** It should be clear what the automatically derived `Show`, `Read` and `Eq` do. But what do you think the order on trees derived with `Ord` should be? *Hint.* This is a non-trivial question. So examine it first for the type of lists. In that case, the automatically derived order is the [lexicographic order](#), which is like the dictionary order.

## Basic functions on binary trees

To get started, let's mirror trees, so that e.g. from the above we get



We do this as follows:

```
mirror :: BT a -> BT a
mirror Empty = Empty
mirror (Fork x l r) = Fork x (mirror r) (mirror l)
```

Running this on the above example we get

```
mirror btxample = Fork 8 (Fork 16 (Fork 20 Empty Empty) Empty) (Fork 4 Empty (Fork 2 Empty Empty))
```

This notation for trees is not very good for visualizing them, as you can see, but is very good for computation.

We define the *size* of a tree as its total number of nodes:

```
size :: BT a -> Integer
size Empty      = 0
size (Fork x l r) = 1 + size l + size r
```


Since we are considering binary trees, the size (i.e., the number of nodes) is also equal to the number of leaves minus one:

```
leaves :: BT a -> Integer
leaves Empty      = 1
leaves (Fork x l r) = leaves l + leaves r
```

We define the *height* of a tree to be the length of the longest path from the root, measured in number of nodes:

```
height :: BT a -> Integer
height Empty      = 0
height (Fork x l r) = 1 + max (height l) (height r)
```

A balanced binary tree has height approximately log of its size, whereas a binary tree which is very unbalanced, such as



```
btleft = Fork 20 (Fork 16 (Fork 8 (Fork 4 (Fork 2 Empty Empty) Empty) Empty) Empty) Empty
```

has height approximately equal to its size.

## Directions, addresses and paths in binary trees

A video discussing the next few sections is [available on Canvas](#).

To pick a subtree of a binary tree, we go left or right successively, until we find it. But a wrong list of directions, called an address here, may be given, and hence we need the `Maybe` type for the output:

```
data Direction = L | R deriving (Show)
type Address   = [Direction]

subtree :: Address -> BT a -> Maybe(BT a)
subtree []      t      = Just t
```

```

subtree ( _: _ ) Empty      = Nothing
subtree ( L:ds ) (Fork _ l _) = subtree ds l
subtree ( R:ds ) (Fork _ _ r) = subtree ds r

```

Following the above pattern, we can define a function that checks whether an address in a given tree is valid:

```

isValid :: Address -> BT a -> Bool
isValid [] _ = True
isValid ( _: _ ) Empty = False
isValid ( L:ds ) (Fork _ l _) = isValid ds l
isValid ( R:ds ) (Fork _ _ r) = isValid ds r

```

The list of valid addresses for subtrees can be computed as follows:

```

validAddresses :: BT a -> [Address]
validAddresses Empty = []
validAddresses (Fork _ l r) = []
                                ++ [L:ds | ds <- validAddresses l]
                                ++ [R:ds | ds <- validAddresses r]

```

List comprehensions can always be eliminated. In this example they become

```

validAddresses' :: BT a -> [Address]
validAddresses' Empty = []
validAddresses' (Fork _ l r) = []
                                ++ (map (L:) (validAddresses' l))
                                ++ (map (R:) (validAddresses' r))

```

We expect that

```
isValid ds t = ds `elem` (validAddresses t)
```

Or, in words, an address is valid if and only if it is an element of the list of valid addresses. Should this be intuitively clear? The statement, yes. But the fact, given our definitions, I don't think so. I would say that it requires a convincing argument. In any case, intuition is something we develop based on convincing arguments we learn.

The list of all paths from the root to a leaf has a similar definition:

```

btpaths :: BT a -> [[a]]
btpaths Empty = []
btpaths (Fork x l r) = [x:xs | xs <- btpaths l]
                        ++ [x:xs | xs <- btpaths r]

```

## Proofs on binary trees by induction

If we have a property  $P$  of trees, and we want to show that  $P(t)$  holds for all trees  $t$ , we can do this by *induction on trees* as follows:

- Argue that  $P(\text{Empty})$  holds.
- Argue that if  $P(l)$  and  $P(r)$  hold for given trees  $l$  and  $r$ , then it holds for  $P(\text{Fork } x \ l \ r)$  where  $x$  is arbitrary.

We are not going to emphasize proofs in this module, but we will indicate when some claims genuinely require proofs, and, moreover, we will try to be precise regarding the specifications of the programs we write.

It is often the case that somebody shows us a clever algorithm and we feel stupid because we don't understand it. But this feeling is wrong. If we don't understand an algorithm, what is missing is a proof. A proof is an explanation. This is what proof means. In order to understand an algorithm we need

- the algorithm itself,
- a precise description of what it is intended to do, and
- a convincing explanation that the algorithm does do what it is intended to do.

Programs alone are not sufficient. We need to know what they are intended to accomplish, and we want to know an explanation justifying that they accomplish what we promise. This promise is called the *specification* of the algorithm / program. Program correctness means "the promise is fulfilled". One way to attempt to prove that the promise is fulfilled is to *test* the program. But actually, all that testing can do it to show that the promise is *not* fulfilled, by finding counterexamples. When the good examples work, we have some kind of evidence that the algorithm works, but not full confidence, because we may have missed examples of inputs that give wrong outputs. Full confidence can only be given by a convincing explanation, also known as *proof*. If you ever asked yourself what "proof" really means, the ultimate answer is "convincing argument".

## Functional proofs

The dependently typed language [Agda](#) allows to write functional programs *and* their correctness proofs, where the [proofs themselves are written as functional programs](#). As an example, [here is a computer-checked proof](#) of the above relation between the functions `isValid` and `validAddresses` in Agda. This is not examinable, and is included here for the sake of illustration only.

## Traversals in binary trees



A video discussing the next few sections is [available on Canvas](#).

We now define the standard in-order and pre-order [traversals](#).

```
treeInOrder :: BT a -> [a]
treeInOrder Empty = []
treeInOrder (Fork x l r) = treeInOrder l ++ [x] ++ treeInOrder r

treePreOrder :: BT a -> [a]
treePreOrder Empty = []
treePreOrder (Fork x l r) = [x] ++ treePreOrder l ++ treePreOrder r
```

For instance, for the trees `btexample` and `btleft` considered above,



we get:

```
> (treeInOrder btexample, treePreOrder btexample)
([2,4,8,16,20],[8,4,2,16,20])
> (treeInOrder btleft, treePreOrder btleft)
([2,4,8,16,20],[20,16,8,4,2])
```

[Breadth-first traversal](#) is trickier. We first define a function that given a tree, produces a lists of lists, with the nodes of level zero (just the root), then the nodes of level one (the successors of the root), then the nodes of level two, and so on:

```
levels :: BT a -> [[a]]
levels Empty = []
levels (Fork x l r) = [[x]] ++ zipappend (levels l) (levels r)
  where
    zipappend [] yss = yss
    zipappend xss [] = xss
    zipappend (xs:xss) (ys:yss) = (xs ++ ys) : zipappend xss yss
```

(Compare `zipappend` to the prelude function [zipWith](#).) For example:

```
> levels btexample
[[8],[4,16],[2,20]]
> levels btleft
[[20],[16],[8],[4],[2]]
```

With this we can define

```
treeBreadthFirst :: BT a -> [a]
```



```
treeBreadthFirst = concat . levels
```

where `.` stands for function composition (look it up in our textbook), and the prelude function `concat :: [[a]] -> [a]` concatenates a list of lists, for example getting `[8,4,16,2,20]` from `[[8],[4,16],[2,20]]`. For further discussion about breadth-first search, see [The under-appreciated unfold](#) (a free version is at the [authors' web page](#)), but this is probably beyond your current level for most of you.

## Inverting traversals (generating trees)

A video discussing the next few sections is [available on Canvas](#).

Many different trees can have the same (in-order/pre-order/breadth-first) traversal, as we saw above with `btexample` and `btleft`, which have the same in-order traversal. In other words, all of the functions

```
treeInOrder, treePreOrder, treeBreadthFirst :: BT a -> [a]
```

are *non-injective* and hence non-invertible. Nonetheless, an interesting and tractable problem is to try to construct a binary tree with a given (in-order/pre-order/breadth-first) traversal, or even to generate *all possible binary trees* with a given traversal.

As an example, the following will produce a *balanced* binary tree given its in-order traversal (which will be a *binary search tree* if the input is sorted):

```
balancedTree :: [a] -> BT a
balancedTree [] = Empty
balancedTree xs = let (ys, x:zs) = splitAt (length xs `div` 2) xs in
  Fork x (balancedTree ys) (balancedTree zs)
```

(The prelude function `splitAt` splits a list in two lists at a given position.) This satisfies the equation

```
treeInOrder (balancedTree xs) = xs
```

for all `xs :: [a]`. In the other direction, it is certainly **not** the case that

```
balancedTree (treeInOrder t) = t
```

for all `t :: Bin a`, for instance

```
balancedTree (treeInOrder btleft) = Fork 8 (Fork 4 (Fork 2 Empty Empty) Empty) (Fork 20 (Fork 16 Empty Empty) Empty)
```

which is not equal to `btleft`. Indeed, the composite function

```
balance :: BT a -> BT a
balance = balancedTree . treeInOrder
```

which applies `treeInOrder` to a tree followed by `balancedTree` to the resulting list can be seen as an operation for rebalancing a binary tree.

Now, using list comprehensions, it is a small step from the function `balancedTree` above to a function generating *all* binary trees with a given in-order traversal.

```
inOrderTree :: [a] -> [BT a]
inOrderTree [] = [Empty]
inOrderTree xs = [Fork x l r | i <- [0..length xs-1],
  let (ys, x:zs) = splitAt i xs,
  l <- inOrderTree ys, r <- inOrderTree zs]
```

This satisfies the property that

```
elem t (inOrderTree xs)
```

if and only if

```
treeInOrder t = xs
```

for all `t :: BT a` and `xs :: [a]`. For example, running

```
> inOrderTree [1..3]
[Fork 1 Empty (Fork 2 Empty (Fork 3 Empty Empty)),Fork 1 Empty (Fork 3 (Fork 2 Empty Empty) Empty),Fork 2 (Fork 1 Empt
```

successfully computes all five binary search trees whose in-order traversal is `[1,2,3]` :

```

  1
 / \
  2
 / \
  3
 / \
Fork 1 Empty (Fork 2 Empty (Fork 3 Empty Empty))

```

```

  1
 / \
  3
 / \
  2
 / \
Fork 1 Empty (Fork 3 (Fork 2 Empty Empty) Empty)

```

```

  2
 / \
 / \
1   3
/ \ / \
Fork 2 (Fork 1 Empty Empty) (Fork 3 Empty Empty)

```

```

  3
 / \
  1
 / \
  2
 / \
Fork 3 (Fork 1 Empty (Fork 2 Empty Empty)) Empty

```

```

  3
 / \
  2
 / \
  1
 / \
Fork 3 (Fork 2 (Fork 1 Empty Empty) Empty) Empty

```

**Task:** write a function `preOrderTree :: [a] -> [BT a]`, with the property that `elem t (preOrderTree xs)` if and only if `treePreOrder t = xs` for all `t :: Bin a` and `xs :: [a]`.

**Very hard task:** write a function `breadthFirstTree :: [a] -> [BT a]`, with the property that `elem t (breadthFirstTree xs)` if and only if `treeBreadthFirst t = xs` for all `t :: Bin a` and `xs :: [a]`. [solution](#)